

IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L.1000

PER IL TUO **Commodore**

Una pubblicazione della J.soft editrice

sped. in abb. post. Gruppo 11/70



PAPER

SOFT

29

Anno 2 - N° 29 - 26 luglio 1985

VIC-20

**La corona
di pietra**

VIC-20

Night duel

Tombola

C16

Supermind

C16

C64

Semaforo

C64

New York

Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono 'parole' racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [< >] deve essere premu-

to contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario" questo dovrà essere premuto contemporaneamente al tasto CONTROL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

[CLR]	SHIFT CLR/HOME		{ CYN }	CTRL 4		[< 7 >]		
[HOME]	CLR/HOME		{ PUR }	CTRL 5		[< 8 >]		
[SU]	SHIFT [] CRSR []		{ GRN }	CTRL 6		{ F1 }		
[GIU']	[] CRSR []		{ BLU }	CTRL 7		{ F2 }		
[SIN]	SHIFT [←] CRSR [→]		{ YEL }	CTRL 8		{ F3 }		
[DES]	[←] CRSR [→]		[< 1 >]			{ F4 }		
[RVS]	CTRL 9		[< 2 >]			{ F5 }		
[OFF]	CTRL 0		[< 3 >]			{ F6 }		
[BLK]	CTRL 1		[< 4 >]			{ F7 }		
[WHT]	CTRL 2		[< 5 >]			{ F8 }		
[RED]	CTRL 3		[< 6 >]					

Per ricevere gli arretrati di **PAPER** compilare il seguente tagliando:



J.soft **Compilare ed inviare in busta chiusa a:**
viale Restelli, 5 - 20124 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____

Desidero ricevere i seguenti numeri arretrati di

PAPER _____ Anno _____

PAPER dedicato a _____ Anno _____

a L. 2.000 cad. per un totale di L. _____ a mezzo

assegno allegato

contanti allegati





VIC-20

7 **La corona di pietra (16K Ram)**
seconda parte
di E. Ceppi

12 **Night duel (joystick)**
di H. Pierce trad. e adatt. di F. Sarcina

C64

23 **Semaforo**
di T. Harrison trad. e adatt. di S. Albarelli

26 **New York**
di K. Eastern trad. e adatt. di P. Maffei

C16-plus/4

15 **Tombola**
di A. Colonna

20 **Supermind**
di B. Kerling trad. e adatt. di S. Albarelli

J.soft s.r.l.
**DIREZIONE, REDAZIONI
E AMMINISTRAZIONE**
Via Restelli, 5
20124 Milano
Tel.: 68.88.228-68.37.97

DIRETTORE RESPONSABILE:
Pietro Dell'Orco

COORDINAMENTO TECNICO:
Mauro Cristuibi Grizzi

REDAZIONE:
Franco Sarcina

GRAFICA E IMPAGINAZIONE
Margherita La Noce
Ivana Rossi
Raffaella Toffolatti

FOTOCOMPOSIZIONE:
Graphotek
Via Astesani, 16 - Milano
Tel. 64.80.397

CONTABILITÀ:
Giulia Pedrazzini
Flavia Bonaiti

**AUTORIZZAZIONE ALLA
PUBBLICITÀ:**
Tribunale di Milano n° 178
del 14.04.1984

STAMPA:
Intergrafica - Pioltello (MI)
Rivista associata
all'Unione stampa
Periodica Italiana



PUBBLICITÀ
Concessionario per l'Italia e l'Estero
J.Advertising s.r.l.
V.le Restelli, 5
20124 MILANO
Tel. (02)
68.82.895-68.80.606-68.87.233
Tlx 316213 REINA I
Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero:
SODIP - Via Zuretti, 25
20125 MILANO
Spedizione in abbonamento
postale Gruppo II/70
Prezzo della rivista L. 1.000
Numero arretrati L. 2.000

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI



Stop Bug

Se non l'hai già fatto, memorizza e salva su nastro o disco la versione di Stop Bug per il tuo computer. Questo programma ti permette, una volta caricato ed attivato con Run, di digitare i listati di Papersoft Commodore senza introdurre errori di battitura.

Come avrai notato, le linee dei nostri programmi terminano con la parola ":rem" seguita da un numero, detto checksum; non copiare questa parte della linea, che serve solo per tua informazione. Quando Stop Bug è in funzione, infatti, tutte le volte che premi il tasto Return dopo aver digitato una linea di programma, appare nell'angolo superiore sinistro dello schermo un numero, che deve coincidere con il checksum da noi pubblicato (sempre che tu non abbia usato

comandi abbreviati, come ? per Print).

Usando Stop Bug tieni conto del fatto che gli spazi non influiscono sui checksum, per cui fai particolare attenzione alle istruzioni Print dove sono spesso presenti. Inoltre eventuali inversioni di caratteri non vengono segnalate.

Un altro metodo per utilizzare Stop Bug, soprattutto se sei abituato ad usare le abbreviazioni dei comandi Basic, è controllare i checksum listando il programma e battendo Return su ogni linea.

Se intendi sospendere il lavoro di digitazione, puoi salvare su nastro o disco il programma che stai memorizzando senza che Stop Bug interferisca; al momento di riprendere il lavoro ricordati però di caricare ed attivare Stop Bug prima di ricaricare il tuo programma.

Per disabilitare Stop Bug premi Run/Stop e Restore (su C64 e VIC 20) oppure Run/Stop e Reset (su C16 e Plus/4). Per riattivarlo batti-

*SYS PEEK(56)*256 per il VIC 20;
SYS 16128 per il C16 e il PLUS/4;
SYS 40704 per il C 64.*

STOP BUG per C64 e VIC 20

```
100 POKE55,0:POKE56,PEEK(56)-1:R=PEEK(56)*256:PRINT"{CLR}
ATTENDERE PREGO..."
110 FORI=RTOR+132:READA:CK=CK+A:IFA=999THENA=R/256
120 POKEI,A:NEXT
130 IFCK<>22689THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
TA":END
140 SYS(R):PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIVAT
O.":NEW
200 DATA173,36,3,201,32,208,1,96,141
210 DATA33,999,173,37,3,141,34,999,169
220 DATA32,141,36,3,169,999,141,37,3
230 DATA169,0,133,254,96,32,87,241,133
240 DATA251,134,252,132,253,8,201,13,240
250 DATA17,201,32,240,5,24,101,254,133
260 DATA254,165,251,166,252,164,253,40,96
270 DATA169,13,32,210,255,165,214,141,251
280 DATA3,206,251,3,169,0,133,216,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,254,169,0
310 DATA133,254,172,33,999,192,87,208,6
320 DATA32,205,189,76,117,999,32,205,221
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,214,76,55,999
```

STOP BUG per C16 e PLUS/4

```
100 POKE55,0:POKE56,63:PRINT"{CLR}ATTENDERE PREGO..."
110 FORI=16128TO16260:READA:CK=CK+A:POKEI,A:NEXT
120 IFCK<>14248THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
TA":END
130 SYS16128:PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIV
ATO.":NEW
200 DATA173,34,3,201,32,208,1,96,141
210 DATA33,63,173,35,3,141,34,63,169
220 DATA32,141,34,3,169,63,141,35,3
230 DATA169,0,133,6,96,32,232,235,133
240 DATA3,134,4,132,5,8,201,13,240
250 DATA17,201,32,240,5,24,101,6,133
260 DATA6,165,3,166,4,164,5,40,96
270 DATA169,13,32,210,255,165,205,141,251
280 DATA3,206,251,3,169,0,133,207,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,6,169,0
310 DATA133,6,172,33,63,192,232,208,6
320 DATA32,95,164,76,117,63,32,7,208
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,205,76,55,63
```

Nessuno è perfetto, ma...

In queste ultime settimane le lettere indirizzate alla nostra Redazione hanno subito un notevole aumento; ciò ci fa senz'altro piacere, in quanto significa indubbiamente che la nostra "rivistina" tascabile sta aumentando la cerchia dei suoi lettori. D'altra parte, però, ci è materialmente impossibile rispondere a tutte le lettere, e in particolare a coloro che ci scrivono segnalando presunti errori nei listati. A questo proposito vogliamo precisare che nei casi in cui abbiamo accertato l'effettiva presenza di errori, questi ultimi sono stati rilevati in appositi "errata corrige" pubblicati nei numeri successivi della rivista. Per fortuna questi casi si sono verificati molto raramente, ed in genere erano dovuti ad inconvenienti in fase di composizione o di stampa, e non ad errori nella logica dei programmi: cogliamo infatti l'occasione per ribadire il fatto che il software pubblicato viene sempre accuratamente "testato" in Redazione prima della stampa dei listati.

Ricordiamo perciò ancora una volta ai nostri lettori che ciò che essi digitano deve coincidere esattamente col listato pubblicato, poiché anche la differenza più insignificante può provocare il malfunzionamento del programma. Prendiamo infatti un caso che si potrebbe definire emblematico: alcuni lettori che avevano digitato il gioco "Il Paracadutista" per C64, comparso sul n. 19, ci hanno scritto segnalando un errore di tipo "ILLEGAL QUANTITY" in una delle linee che vanno da 4200 alla fine del programma. Dopo aver verificato l'esattezza del listato, abbiamo provato a risalire alla possibile fonte dell'inconveniente; abbiamo scoperto allora che, con ogni probabilità, tutto ciò era dovuto ad un errore di battitura commesso dai suddetti lettori nel copiare le istruzioni DATA che si trovano nella parte iniziale del listato. In particolare tale errore poteva essere (tipicamente!) l'aver scambiato la lettera O con lo zero.

Per fortuna, comunque, adesso c'è STOP BUG, e siamo certi che casi come questo diverranno molto più rari; il nostro augurio sincero è che tutti i programmi che copierete da Papersoft funzionino sempre al primo colpo!

Vogliamo inoltre approfittare di questo spazio per esortare i lettori ad inviarci i programmi che scrivono; abbiamo già pubblicato molti lavori tra quelli che ci sono stati spediti e garantiamo in ogni caso una risposta a tutti coloro che ci faranno pervenire il loro software.

Per concludere, diamo a tutti gli amici che usano il C64 con il registratore una notizia importante: nei prossimi due numeri troverete SCHEDARIO, un potente e versatile data-base che vi consentirà di memorizzare su cassetta archivi di ogni genere!

La Redazione

La Corona di Pietra

Seconda parte

Ecco la seconda (ed ultima) parte del fantastico adventure game per VIC 20 di Papersoft Commodore. La prima puntata è stata pubblicata sul numero della settimana scorsa; se ti manca corri in edicola a comprarlo e, se non lo trovi, spedisce il coupon di richiesta numeri arretrati.

Nel caso invece tu abbia già digitato e salvato la prima parte de "La

Corona di Pietra", carica in memoria dopo aver attivato STOP BUG; puoi quindi iniziare a battere il listato di questa settimana. Quando hai finito, salva il tutto: ora sei pronto per partire alla ricerca della Corona! Per le spiegazioni sul funzionamento del gioco ti rimandiamo all'articolo pubblicato sullo scorso numero.

```

1620 IFSE%=24ANDSE%(22)=-10ANDOD$(2)="TECA"THENPRINT"DENT
    RO C'E'LA CORONA":GOTO410                :rem 108
1630 IF(OD$(2)="TECA"ANDSE%<>24)OR(OD$(2)="TORRE"ANDSE%<>
    25)THENPRINT"QUALE "OD$(2)" ?":GOTO410  :rem 217
1640 IFOD$(2)="CROCEFISSO"THENOD$(2)="CROCIFISSO"
                                           :rem 227
1650 IFOD$(2)="CROCIFISSO"ANDSE%<>21THENPRINT"QUALE "OD$(
    2)" ?":GOTO410                          :rem 180
1660 IFSE%<>21OROD$(2)<>"CROCIFISSO"THENPRINT"NON NOTO NU
    LLA DI SPECIALE":GOTO410                :rem 5
1670 PRINT"IL BRACCIO DESTRO SI{ 2 SPAZI}PUO'RUOTARE!!":G
    OTO410                                    :rem 27
1680 IFSE%<>21OROD$(2)<>"BRACCIO"OROD$(3)="SINISTRO"THENP
    RINT"COSA???" :GOTO780                  :rem 81
1690 PRINT"IL BACCIO SI RUOTA E..IL MURO SI APRE!!!":E%=2
    3:BR%=1:GOTO410                          :rem 77
1700 IFOD$(2)<>"MANOSCRITTO"THENPRINT"COSA???" :GOTO780
                                           :rem 151
1710 IFSE%(10)<>-1THENPRINT"QUALE MANOSCRITTO ?":GOTO410
                                           :rem 177
1720 PRINT"DICE:LA CORONA DI PI =ETRA E'NEL CASTELLO INCA
    NTATO.":GOTO410                          :rem 65
1730 IFOD$(2)<>"PONTE"THENPRINT"COSA???" :GOTO780:rem 205
1740 IFSE%<>3ANDSE%<>4ANDSE%<>28ANDSE%<>29THENPRINT"NON V
    EDO NESSUN PONTE QUI.":GOTO410         :rem 140
1750 IFSE%=28THENPRINT"NON POSSO APRIRLO DAL=L'ESTERNO.":
    GOTO410                                   :rem 47
1760 IFSE%=29THENPRINT"O.K.":E%=28:PL%=1:FORT=0TO750:NEXT
    T:GOTO90                                  :rem 54
1770 PRINT"NON ESSERE SCIOCCO!!!":GOTO780  :rem 202

```

```

1780 IFOD$(2)="TOPO"OROD$(2)="TOPI"OROD$(2)="STREGA"OROD$
      (2)="FANTASMA"THEN910 :rem 134
1790 IFOD$(2)<>"CANE"THENPRINT"VUOI SCHERZARE,VERO?":GOT
      O410 :rem 8
1800 IFSE%(7)<>-1THEN910 :rem 62
1810 PRINT"O.K."CHR$(13)"FUNZIONA!IL CANE SE NEVA!!!":CA%
      =2:N%=18:SE%(21)=-10:GOTO410 :rem 10
1820 IFOD$(2)="TOPI"OROD$(2)="TOPO"THEN1920 :rem 234
1830 IFSE%(4)<>-1ANDSE%(6)<>-1ANDSE%(9)<>-1THEN910
      :rem 31
1840 IFOD$(2)="FANTASMA"THENPRINT"MA,SECONDO TE,COME FACC
      IO ???":GOTO410 :rem 122
1850 IFOD$(2)="STREGA"THENPRINT"MI E'IMPOSSIBILE.":GOTO41
      0 :rem 86
1860 IFOD$(2)<>"CANE"THENPRINT"COSA????":GOTO780:rem 161
1870 IFSE%(4)<>-1ANDSE%(6)=-1ANDSE%(9)<>-1THENPRINT"IL CO
      LTELLO NON SERVE A MOLTO.":GOTO410 :rem 130
1880 IFSE%(9)=-1THENPRINT"O.K.":GOTO1910 :rem 118
1890 IFSE%(4)=-1ANDSE%(8)=-1THENPRINT"BANG,BANGGG!!":GOTO
      1910 :rem 92
1900 PRINT"IL FUCILE E'SCARICO!!":GOTO410 :rem 162
1910 NOS(21)="CANE MORTO":CA%=2:N%=18:GOTO410 :rem 22
1920 IFSE%<>21THENPRINT"QUALI TOPI?":GOTO410 :rem 244
1930 IFSE%(0)<>-1THEN910 :rem 59
1940 IFNOS(14)="TOPI MORTI"THENPRINT"MA SONO GIA'MORTI!!"
      :GOTO410 :rem 182
1950 PRINT"O.K.":NOS(14)="TOPI MORTI":N%=22:SE%(0)=21:NO%
      =NO%-1:GOTO410 :rem 86
1960 IFSE%<>26ORSE%(24)<>26THENPRINT"TI AVEVO CONSIGLIATO
      { 2 SPAZI}DI DESISTERE!!":GOTO410 :rem 206
1970 IFSE%(6)=-1ORSE%(9)=-1ORSE%(26)=-1THENPRINT"TRA LE C
      OSE CHE HO CI E'QUELLA GIUSTA":GOTO410 :rem 14
1980 PRINT"NON HO NULLA CON ME{ 3 SPAZI}CHE MI POSSA TRAR
      RE DAQUESTA SITUAZIONE.":GOTO3050 :rem 155
1990 CO%=0:IFOD$(2)="STREGA"THENPRINT"LA STREGA MI TRASFO
      RMAIN UN TOPO!!":GOTO3050 :rem 210
2000 FORI=0TO27:IFOD$(2)=LEFT$(NOS(I),LEN(OD$(2)))ANDSE%(
      I)=SE%THENCO%=1 :rem 180
2010 NEXT :rem 2
2020 IFCO%=0THENPRINT"COSA????":GOTO780 :rem 108
2030 IFOD$(2)="CANE"THENCA%=1:GOTO360 :rem 120
2040 IFOD$(2)="FANTASMA"THENPRINT"NON ESSERE SCIOCCO!!":G
      OTO410 :rem 90
2050 IFOD$(2)="TOPO"OROD$(2)="TOPI"THENPRINT"NON DIRE SCI
      OCCHEZZE!":GOTO410 :rem 93
2060 IFSE%<>18ORBO=1OROD$(2)<>"BOTTE"THENPRINT"O.K.":GOTO
      410 :rem 172
2070 IFSE%<>SE%(12)THENPRINT"QUALE BOTTE?":FORT=0TO750:N
      EXTT:GOTO90 :rem 255
2080 PRINT"O.K.":PRINT"DIETRO LA BOTTE C'E'UNPASSAGGIO!!"
      :BO=1:N%=19: :rem 142

```



```

2090 FORT=0TO1500:NEXTT:GOTO90           :rem 133
2100 P0$="SONO IN UNA SPIAGGIA. A NORD C'E'IL MARE,A
      { 2 SPAZI}SUD LA FORESTA,A EST{ 2 SPAZI}" :rem 108
2110 P1$="IL CASTELLO,A OVEST LASPIAGGIA.           :rem 178
2120 S%=7:E%=1:O%=0:RETURN                 :rem 248
2130 P0$="SONO NEL TUO CASTELLO.IL MARE E'A NORD,LA FORES
      TA A SUD,IL PROMON=                       :rem 209
2140 P1$="TORIO A EST E LA SPIAGGIA A OVEST.       :rem 156
2150 S%=8:E%=2:O%=0:IFSE%(22)=-1THENPRINT"{CLR}"P0$;P1$;P
      2$:GOTO3070                               :rem 238
2160 RETURN                                 :rem 169
2170 P0$="SONO SUL PROMONTORIO: IL MARE E'A NORD,LA FORES
      TA A SUD,IL PONTE{ 2 SPAZI}"             :rem 159
2180 P1$=P0$+" E' A EST E IL CASTELLO A OVEST.     :rem 135
2190 S%=9:E%=3:O%=1:RETURN                 :rem 4
2200 P0$="SONO AD OVEST DEL PON=TE "+P0$+" .IL MA=RE E'A
      NORD,LA FORESTA                           :rem 227.
2210 P1$="E'A SUD,IL PROMONTORIOE'A OVEST.         :rem 247
2220 S%=10:O%=2:IFCR%=0ORTR%=1THENE%=4          :rem 205
2230 RETURN                                 :rem 167
2240 P0$="SONO SUL LATO EST DEL PONTE "+P0$+":A NORD C'E'
      IL MARE,A{ 2 SPAZI}"                     :rem 159
2250 P1$="SUD UN CREPACCIO,A EST LA BAIA.          :rem 6
2260 IFCR%=0THENP2$=CHR$(13)+"IL PONTE CROLLA DIETRODI ME
      !!!":CR%=1:P0$="{ 2 SPAZI}CROLLATO "      :rem 24
2270 E%=5:IFTR%=1THENO%=3                   :rem 188
2280 RETURN                                 :rem 172
2290 P0$="SONO IN UNA BAIA. A{ 3 SPAZI}NORD C'E'UN BRACCI
      O DIMARE ED OLTRE LA SPIAG              :rem 103
2300 P1$="GIA DI UN'ISOLA.LA FO=RESTA E'A SUD,LA SPIAGGIA
      AD EST ED IL PONTE                       :rem 33
2310 P2$="CROLLATO A OVEST.                    :rem 43
2320 S%=11:E%=6:O%=4:RETURN                 :rem 46
2330 P0$="SONO SULLA SPIAGGIA.{ 2 SPAZI}IL MARE E'A NORD,
      LA FORESTA A SUD,LA SPIAG =             :rem 252
2340 P1$="GIA E'A EST E LA BAIA A OVEST.         :rem 22
2350 S%=12:E%=6:O%=5:RETURN                 :rem 51
2360 P0$="SONO NELLA FORESTA.LA SPIAGGIA E'A NORD,LA
      { 2 SPAZI}FORESTA A SUD,OVEST , EST.     :rem 106
2370 N%=0:S%=7:E%=8:O%=7:RETURN             :rem 39
2380 P0$="SONO NELLA FORESTA.IL CASTELLO E'A NORD,LA
      { 2 SPAZI}FORESTA A SUD,OVEST , EST.     :rem 134
2390 N%=1:S%=8:E%=9:O%=7:RETURN             :rem 44
2400 P0$="SONO NELLA FORESTA.IL PROMONTORIO E'A NORD, LA
      FORESTA A SUD,EST, OVEST.               :rem 144
2410 N%=2:S%=9:E%=10:O%=8:RETURN            :rem 80
2420 P0$="SONO NELLA FORESTA.IL PONTE "+P0$+" E'A NORD,LA
      FORESTA E'A{ 3 SPAZI}"                 :rem 172
2430 P1$="SUD ED OVEST,IL CREPACCIO E'A EST.     :rem 228
2440 N%=3:S%=10:O%=9:RETURN                 :rem 59

```

2450 P0\$="SONO NELLA FORESTA.LA BAI A E'A NORD,LA FORESTA
E'A SUD E A EST, IL :rem 206
2460 P1\$="CREPACCIO E'A OVEST. :rem 197
2470 N%=5:S%=11:E%=12:RETURN :rem 97
2480 P0\$="SONO NELLA FORESTA.LASPIAGGIA E'A NORD,LA FORES
TA A SUD,EST,OVEST.{SIN} :rem 10
2490 N%=6:S%=12:E%=12:O%=11:RETURN :rem 178
2500 P0\$="SONO SULLA SPIAGGIA{ 3 SPAZI}DELL'ISOLA.LA FORE
STA E'A NORD,IL MARE A SUD :rem 81
2510 P1\$="E A EST,C'E'UN CUNICO=LO A OVEST.A SUD,OLTREIL
MARE,C'E'LA BAI A. :rem 223
2520 N%=16:O%=14:RETURN :rem 74
2530 P0\$="SONO IN UN CUNICOLO.A NORD C'E'UNA GROTTA,A EST
LA SPIAGGIA,E AD{ 2 SPAZI}" :rem 18
2540 P1\$="OVEST UN'ALTRO CUNICO=LO. :rem 115
2550 N%=17:E%=13:O%=15:RETURN :rem 148
2560 P0\$="SONO IN UN CUNICOLO.A NORD C'E'UNA CAVERNA,AEST
UN'ALTRO CUNICOLO. :rem 48
2570 N%=18:E%=14:IFCA%<>2THENN%=-1 :rem 136
2580 RETURN :rem 175
2590 P0\$="SONO IN UNA FORESTA.A NORD E A EST C'E'UN
{ 3 SPAZI}PRATO,A SUD UNA SPIAG=GIA. :rem 161
2600 N%=26:S%=13:E%=25:RETURN :rem 149
2610 P0\$="SONO IN UNA GROTTA.LA USCITA E'A SUD. :rem 162
2620 S%=14:RETURN :rem 254
2630 P0\$="SONO IN UN CUNICOLO.LAUSCITA E' A SUD.":S%=15:I
FBO=0THENRETURN :rem 59
2640 P0\$="SONO IN UN CUNICOLO.CISONO DUE CUNICOLI A
{ 3 SPAZI}NORD E A SUD. :rem 229
2650 N%=19:S%=15:RETURN :rem 86
2660 P0\$="SONO IN UN CUNICOLO:A SUD C'E'UN CUNICOLO,A OVE
ST LA CANTINA DEL{ 2 SPAZI}" :rem 98
2670 P1\$="CASTELLO INCANTATO. :rem 249
2680 S%=18:O%=20:RETURN :rem 85
2690 P0\$="SONO NELLA CANTINA DELCASTELLO INCANTATO.A
{ 2 SPAZI}NORD C'E'UN CORRIDOIO, :rem 59
2700 P1\$="A EST UN CUNICOLO. :rem 39
2710 N%=21:E%=19:RETURN :rem 66
2720 P0\$="SONO IN UN CORRIDOIO. C'E'UNA CANTINA A SUD E U
NA STANZA A NORD :rem 114
2730 P1\$=".{ 2 SPAZI}SUL MURO EST C'E'FISSATO UN CROCIFIS
SO. :rem 213
2740 S%=20:IFNO\$(14)="TOPI MORTI"THENN%=22 :rem 175
2750 IFBR%=1THENP1\$=" E A EST.":E%=23 :rem 191
2760 RETURN :rem 175
2770 P0\$="SONO IN UNA STANZA CONUNA PORTA CHIUSA.L'U =SCI
TA E'A SUD. :rem 211
2780 S%=21:RETURN :rem 3
2790 P0\$="SONO IN UNA STANZA.LA USCITA E'A OVEST.
{ 5 SPAZI}A NORD C'E'UNO SCIVOLO :rem 171
2800 P1\$="MA E'BLOCCATO DA UN{ 3 SPAZI}FANTASMA INCATENAT

```

O. :rem 150
2810 IFSE%(25)=-10THENN%=24:P1$="CHE PORTA SOTTO. :rem 58
2820 O%=21:RETURN :rem 250
2830 P0$="SONO NELLA STANZA DEL=LA CORONA.A EST C'E'LAARM :rem 154
ERIA.
2840 E%=30:RETURN :rem 242
2850 P0$="SONO IN UN PRATO.A{ 4 SPAZI}NORD C'E'LA TORRE,S :rem 152
EN=ZA PORTE,DEL CACCIA-
2860 P1$="STREGHE;A SUD C'E'LA{ 2 SPAZI}FORESTA.:rem 243
2870 O%=16:RETURN :rem 3
2880 P0$="SONO IN UN PRATO.{ 3 SPAZI}A NORD C'E'UNA RADUR :rem 23
A,A SUD LA FORESTA.
2890 P1$="{ 7 SPAZI}SONO CADUTO IN UNA{ 4 SPAZI}TRAPPOLA :rem 236
DI CORDA!!
2900 IFSE%(24)<>26THENP1$="":N%=28:S%=16 :rem 175
2910 RETURN :rem 172
2920 P0$=" SONO SULLA TORRE DEL{ 2 SPAZI}CACCIA-STREGHE.A :rem 76
SUD{ 2 SPAZI}C'E'UN PRATO.
2930 GI%=25:RETURN :rem 65
2940 P0$="SONO IN UNA RADURA. ILPRATO E'A SUD,IL CA = STE :rem 78
LLO INCANTATO A OVE
2950 P1$="ST:IL PONTE LEVATOIO{ 2 SPAZI}E'":P2$="CHIUSO. :rem 249
2960 IFPL%=1THENO%=29:P2$="APERTO. :rem 18
2970 S%=26:RETURN :rem 9
2980 P0$="SONO NEL CORTILE DEL{ 2 SPAZI}CASTELLO INCANTAT :rem 200
O.LA ARMERIA E'A NORD,A EST
2990 P1$="C'E'IL PONTE LEVATOIO ":P2$="CHIUSO. :rem 134
3000 IFPL%=1THENE%=28:P2$="APERTO. :rem 249
3010 N%=30:RETURN :rem 241
3020 P0$="SONO NELL'ARMERIA PIE=NA DI POLVERE DA SPARO;A :rem 201
SUD C'E'IL CORTILE,
3030 P1$="A OVEST LA STANZA DEL=LA CORONA. :rem 137
3040 S%=29:O%=24:RETURN :rem 82
3050 PRINT"TI AVEVO CONSIGLIATO{ 2 SPAZI}DI NON CIMENTART :rem 139
I IN{ 2 SPAZI}QUESTA IMPRESA;BEN TI STA!!
3060 GOTO3080 :rem 205
3070 PRINT"{GIU'}{RVS}BRAVISSIMO!!!{OFF}HAI RECU=PERATO L :rem 218
A CORONA ED IL TRONO!!
3080 PRINT"{GIU'}{ 3 SPAZI}{RVS}RIGIOCHI [S/N]?:rem 178
3090 GETA$:IFA$<>"S"ANDA$<>"N"THEN3090 :rem 143
3100 PRINT:IFA$="S"THENRUN :rem 119
3110 END:END:END:END:END :rem 224
3120 PRINT"{CLR}{ 8 SPAZI}{RVS}[[<*>]E[<*>]E[<*>]E":PRINT" :rem 201
{ 8 SPAZI}[[<*>]{RVS}{ 4 SPAZI}{OFF}E":PRINT"{RVS} LA
{ 2 SPAZI}CORONA DI PIETRA "
3130 PRINT"{GIU'}IL TUO INFIDO CONSIGLIERE TI HA SOTTRATT :rem 186
O LACORONA DI PIETRA ED IL
3140 PRINT"{SU}TRONO.PER RIOTTENERLO DEVI RITROVARE LA CO :rem 115
RONA.PARTI DAL TUO CA =

```

```

3150 \PRINT"STELLO ALLA SUA RICER=CA,TENENDO PRESENTE
      { 3 SPAZI}CHE DOVRAI RITORNARVI           :rem 200
3160 PRINT"SALVO E CON LA CORONA!IO SARO'PER TE OCCHI EOR
      ECCHIE,VEDI DI NON                         :rem 192
3170 PRINT"FARMI MORIRE.":PRINT"SE IO FOSSI IN TE RI =NUN
      CEREI!!                                     :rem 71
3180 FORI=0TO27:READAR$(I),NO$(I),SE$(I):NEXT    :rem 159
3190 DATADEL,VELENO,19,UN,REMO,7,UN,REMO,2,UNA,BARCA,5,UN
      ,FUCILE,25                                  :rem 160
3200 DATAUNA,TORCIA,16,UN,COLTELLO,11,UN,OSSO,17,DELLE,MU
      NIZIONI,14                                  :rem 15
3210 DATAUNA,SCURE,12,UN,MANOSCRITTO,18,UN,TESCHIO,18,UNA
      ,BOTTE,18                                   :rem 210
3220 DATAUNA,CHIAVE,18,DEI,TOPI,21,DELL',AGLIO,27,UNA,SCA
      LA,14,UNA,BOTTIGLIA,0                       :rem 52
3230 DATADELLE,CAROTE,27,DELLE,MELE,27,DELLE,PERE,27,UN,C
      ANE,15                                       :rem 96
3240 DATA,ALTA,CORONA DI PIETRA,-10,UNA,TECA DI CRISTALLO,24
      ,UNA,TRAPPOLA,26                            :rem 194
3250 DATAUN,FANTASMA INCATENATO,23,UN,TRONCHESE,13,UN,TRO
      NCO,-10                                     :rem 53
3260 PRINT"{GIU'}{ 4 SPAZI}{RVS}PREMI UN TASTO  :rem 72
3270 GETA$:IFA$=""THEN3270                       :rem 187
3280 PO$="PERICOLANTE":RETURN                    :rem 97
3290 PRINT"NON VEDO NIENTE:E'PERICOLOSO MUOVERSI ALLA
      { 2 SPAZI}CIECA!":T0%=T0%-1               :rem 81
3300 IFT0%<-1THENPRINT"SONO CADUTO E MI SONO ROTTO LA TES
      TA!!":GOTO3050                              :rem 212
3310 P0$="":P1$="":P2$=""                        :rem 173
3320 GOTO320                                      :rem 150

```

Night Duel



Joystick

Fai parte di una squadriglia da caccia e stai compiendo una missione notturna nel cielo nemico; in seguito a un'imboscata ti trovi isolato e perdi il contatto visivo coi tuoi compagni. Ora devi cavertela da solo! Gli aerei nemici si dirigono verso di te ad uno ad uno, e tu devi abatterli prima che arrivino abbastanza vicini da colpirti a loro

volta. Puoi vedere sullo schermo il caccia nemico che si avvicina zigzagando e, al centro, il mirino del tuo dispositivo lanciamissili. Usa il joystick per modificare la tua rotta centrando il bersaglio, e il pulsante per far fuoco; tieni presente che ciò che muovi non è il mirino (il quale resta sempre al centro dello schermo) ma bensì l'intera immagi-

ne visualizzata. Appena abbatti un aereo, ne compare subito un altro che si avvicina più rapida-

mente del precedente. La partita termina se vieni colpito.

```

10 REM NIGHT DUEL :rem 233
100 POKE36879,8:PRINT"{CLR}{WHT}{RVS}{ 2 SPAZI}*
    { 2 SPAZI}NIGHT{ 3 SPAZI}DUEL{ 2 SPAZI}*{ 2 SPAZI}"
    :rem 226
105 PRINT"{ 3 GIU'}{ 3 SPAZI}ABBATTI{ 2 SPAZI}L'AEREO"
    :rem 112
110 PRINT"{ 3 SPAZI}NEMICO PRIMA CHE" :rem 103
115 PRINT"{ 4 SPAZI}TI RAGGIUNGA!!" :rem 220
120 PRINT"{ 3 GIU'}{ 3 SPAZI}USA{ 2 SPAZI}IL JOYSTICK"
    :rem 133
130 PRINT"{ 5 SPAZI}PER CENTRARE" :rem 160
135 PRINT"{ 3 SPAZI}L'OBIETTIVO{ 2 SPAZI}NEL" :rem 113
140 PRINT" MIRINO E PER SPARARE" :rem 110
150 PRINT"{ 3 GIU'}{RVS} PULSANTE PER GIOCARE " :rem 249
160 GOSUB450:IFFI=0THEN160 :rem 59
165 I=1:US=7680:UC=38400:POKE36879,8 :rem 51
170 DIMC(11):DIMD$(9) :rem 110
175 D$(9)="." :D$(8)="-" :D$(7)=" [<E>]" :D$(6)=" [<@>]L" :D$(5)
    )="{SIN}*O*{SU}{ 2 SIN}[<R>]" :rem 31
180 D$(4)=" {SIN}*()*{SU}{ 3 SIN}[<T>]O" :D$(3)=" { 3 SIN}
    { 2 *}(O){ 2 *}{SU}{ 5 SIN}U[<E>]I{SU}{ 3 SIN}*[<R>]
    *" :rem 239
185 D$(2)=" { 5 SIN}{ 4 *}( <> ){ 4 *}{SU}{ 8 SIN}N[< 2 +>]
    M{SU}{ 2 SIN}[<G>]{SU}{ 3 SIN}[< 4 O>]" :rem 2
190 D$(1)=" { 9 SIN}[< 4 P>][< 2 O>][< 2 I>]( <> ) [< 2 I>]
    [< 2 O>][< 4 P>]{SU}{ 12 SIN}N[< 2 +>]M{SU}{ 2 SIN}
    [<H>]{SU}{ 3 SIN}[< 4 @>]" :rem 132
195 D$(0)=" { 9 SIN}{ 20 SPAZI}{SU}{ 14 SIN}{ 6 SPAZI}{SU}
    { 4 SIN}{ 4 SPAZI}{SU}{ 5 SIN}{ 7 SPAZI}" :rem 92
200 EX$="{ 10 DES}[< 3 +>]{GIU'}{ 4 SIN}[< 5 +>]{ 9 SIN}
    [< 8 +>]{GIU'}{ 4 SIN}[< 5 +>]{GIU'}{ 4 SIN}[< 2 +>]
    " :rem 244
205 FORX=1TO8:READR(X):NEXT :rem 233
210 FORX=1TO11:READC(X):NEXT :rem 0
215 PRINT"{CLR}{WHT}" :rem 1
220 X=6:Y=10:Q=Y:P=X:K=80:GOSUB305 :rem 76
225 W=X:Z=Y:K=K-I*.2:IFK=<0THEN420 :rem 62
230 GOSUB450:IFJY=156ANDFI=0THEN300 :rem 132
235 IFFI=1THEN345 :rem 241
240 IFINT(RND(1)*20+1)<ITHEN300 :rem 246
245 IFJY=140THENW=X+1:GOTO290 :rem 36
250 IFJY=28THENW=X-1:GOTO290 :rem 247
255 IFJY=152THENZ=Y+1:GOTO290 :rem 44
260 IFJY=148THENZ=Y-1:GOTO290 :rem 47
265 IFJY=136THENW=X+1:Z=Y+1:GOTO290 :rem 177

```

```

270 IFJY=132THENW=X+1:Z=Y-1:GOTO290           :rem 171
275 IFJY=24THENW=X-1:Z=Y+1:GOTO290           :rem 128
280 IFJY=20THENW=X-1:Z=Y-1:GOTO290           :rem 122
285 GOTO230                                     :rem 109
290 IFW<5ORW>15ORZ<3ORZ>21THENW=X:Z=Y:GOTO300 :rem 223
295 X=W:Y=Z:GOSUB305:GOTO225                 :rem 20
300 GOSUB325:JY=R(INT(RND(1)*8+1)):GOTO245    :rem 216
305 POKE781,Q:POKE782,P:SYS58636             :rem 175
310 PRINTD$(0)                                :rem 10
315 POKE781,Y:POKE782,X:SYS58636            :rem 192
320 PRINTD$(INT(K/10)+1):P=X:Q=Y:RETURN      :rem 168
325 POKEUC+209,3:POKEUC+251,3:POKEUC+254,3:POKEUC+297,3
                                                :rem 252
330 POKEUS+209,101:POKEUS+251,64:POKEUS+254,64:POKEUS+297
    ,101:RETURN                                :rem 126
335 DATA140,28,152,148,136,132,24,20       :rem 21
340 DATA102,102,102,90,90,90,90,46,46,46,46 :rem 110
345 S=8142:E=8163:R=38862:D=38883          :rem 38
350 GOSUB325:POKE36878,15                   :rem 189
355 FORF=1TO11                                :rem 61
360 POKES+21,32:POKEE+23,32                 :rem 9
365 POKES,C(F):POKEE,C(F)                   :rem 218
370 POKER,2:POKED,2                          :rem 132
375 S=S-21:E=E-23:R=R-21:D=D-23            :rem 225
380 POKE36875,255-INT(11*F):NEXT:POKE36875,0:POKE36878,0
                                                :rem 250
385 IFX>8ANDX<12ANDY>9ANDY<13THEN395       :rem 6
390 PRINT"{CLR}":GOSUB305:GOTO300           :rem 88
395 PRINT"{CLR}":POKE36879,42:GOSUB415:POKE36878,15:FORQ=
    1TO80:NEXT:POKE36879,8                   :rem 178
400 FORT=255TO128STEP-1:POKE36877,T:NEXT    :rem 241
405 POKE36877,0:POKE36878,0                 :rem 8
410 I=I+1:GOTO215                            :rem 203
415 PRINT"{ 9 GIU'}"EX$:RETURN              :rem 223
420 FORQ=1TO5:POKE36879,25:FORA=1TO30:NEXT:POKE36879,42:F
    ORA=1TO20:NEXT:NEXT                       :rem 67
425 PRINT"{CLR}{ 3 GIU' }{ 6 DES}SEI MORTO!!{ 3 GIU' }":PRI
    NTI-1;"NEMICI ABBATTUTI{ 3 GIU' }"      :rem 60
430 FORQ=1TO10:GETA$:NEXT                    :rem 57
435 INPUT"PROVI ANCORA (S/N)";H$            :rem 124
440 IFH$="S"THENRUN                          :rem 139
445 PRINT"{CLR}":END                        :rem 18
450 FI=0:JY=PEEK(37151):IF(JYAND32)=0THENFI=1 :rem 180
455 JY=JYAND28:POKE37154,127:PP=PEEK(37152)AND128:POKE371
    54,255                                    :rem 195
460 JY=JY+PP:RETURN                          :rem 2

```

Tombola

Ecco uno dei più vecchi passatempi del mondo in versione computerizzata. Il C16 sostituisce colui che normalmente prende in mano le redini del gioco estraendo i numeri e chiamandoli ad alta voce. Il programma ha memorizzate 24 cartelle, per cui i giocatori possono essere al massimo 24. All'inizio si deve introdurre il numero e i nomi dei concorrenti; viene poi chiesto ad ognuno quante quante cartelle desidera (fino a 3) e quali. Ogni cartella ha infatti un "numero di matricola", e quelle disponibili sono contrassegnate dalle matricole 25-36 e 61-72. Chi volesse "personalizzare" il gioco può modificare le linee 1002-1048, ognuna delle quali contiene un'istruzione DATA relativa ad una cartella: il primo dato è la matricola della cartella, seguono i 5 numeri della prima riga, i 5 della seconda e i 5 della terza. In seguito bisogna comunicare al

computer la puntata (in lire) per una singola cartella, in modo che possa venire visualizzato l'importo che ogni giocatore deve pagare prima dell'inizio del gioco. È necessario infine specificare se i premi devono partire dall'ambo, dal terno, ecc. dopo di che inizia l'estrazione dei numeri. Il programma provvede a controllare ogni volta tutte le cartelle e segnala eventuali vincite, visualizzando il nome del concorrente e la matricola della cartella. In questo caso il gioco si ferma e viene emesso un segnale acustico; per proseguire premere "C". L'estrazione può essere in qualsiasi momento sospesa usando il tasto "S" e fatta riprendere con "C". Premendo quest'ultimo tasto dopo che è stata segnalata la tombola, viene infine mostrata la spartizione del "montepremi" tra i vari giocatori.

```

0 REM *****:rem 172
2 REM * TOMBOLA PER C16 - PLUS/4 * :rem 221
4 REM * DI * :rem 249
6 REM * ANTONIO COLONNA * :rem 144
10 REM*****:rem 221
11 COLOR0,3+0,4:COLOR4,3+0,4 :rem 50
12 PRINT" {CLR} {WHT}ATTENDERE.....PREGO !" :rem 236
14 REM (INIZIALIZZAZIONE VARIABILI) :rem 0
16 CLR :rem 72
18 DIMNR(384),CA(72),NC(90),N$(24),V$(99),F1(24),OC%(72) :rem 105
19 VOL8:Y=5 :rem 152
20 H$(2)="AMBO{ 4 SPAZI}":H$(3)="TERNA{ 3 SPAZI}":H$(4)="
   QUATERNA":H$(5)="CINQUINA" :rem 40

```

```

22 H$(6)="TOMBOLA " :rem 230
24 FORI=1TO384:READNR(I):NEXT :rem 79
26 S$="{HOME}{ 23 GIU' }" :rem 250
28 REM (IMMISSIONE DATI) :rem 190
30 GOSUB256 :rem 128
32 INPUT"{HOME}{ 12 GIU' }{ 3 SPAZI}- NUMERO GIOCATORI IN
GARA ";G :rem 240
34 IFG<1ORG>24THEN32 :rem 202
36 FORI=1TOG:PRINT"{HOME}{ 17 GIU' }{ 3 SPAZI}- NOME GIOCA
TORE "; :rem 100
38 PRINT"{ 15 SPAZI}{ 15 SIN}"I;:INPUTN$(I):NEXT:rem 66
40 FORI=1TOG :rem 236
42 GOSUB256:PRINT"{HOME}{ 9 GIU' }{ 3 SPAZI}- ABBINAMENTO
GIOCATORI{ 2 SPAZI}CARTELLE" :rem 116
44 F=0 :rem 27
46 PRINT"{ 2 GIU' } - QUANTE CARTELLE VUOI, "N$(I)" ";:INP
UTF :rem 174
48 IFF<1ORF>3THEN42 :rem 155
50 TC=TC+F:IFTC>24THEN92 :rem 223
52 FORT=1TOF:F1(I)=F:V=V+1 :rem 70
54 PRINT"{ 2 GIU' } "T"{SIN}) CARTELLA NUMERO ";:INPUTCA(V
) :rem 8
55 IFCA(V)>0ANDCA(V)<73THENIFOC%(CA(V))=1THEN62 :rem 44
56 FORT1=1TO369STEP16 :rem 45
58 IFCA(V)=NR(T1)THENV$(CA(V))="SV":OC%(CA(V))=-1:T1=369:
NEXTT1:GOTO64 :rem 170
60 NEXTT1 :rem 42
62 CA(V)=0:V=V-1:GOTO42 :rem 90
64 NEXTT :rem 253
66 IFF=1THENV=V+2:GOTO70 :rem 254
68 IFF=2THENV=V+1 :rem 38
70 FORII=1TO72:OC%(II)=ABS(OC%(II)):NEXTII:NEXTI:rem 249
72 GOSUB256 :rem 134
74 PRINT"{HOME}{ 11 GIU' }{ 3 SPAZI}- PUNTATA PER OGNI CAR
TELLA " :rem 176
76 INPUT"{ 2 GIU' }{ 5 SPAZI}LIRE ";Q :rem 27
78 FORI=1TOG :rem 247
80 PP=F1(I)*Q :rem 209
82 PT=PT+PP :rem 186
84 PRINT"{HOME}{ 17 GIU' }{ 39 SPAZI}" :rem 113
86 PRINT"{HOME}{ 17 GIU' }{ 3 SPAZI}- PUNTATA DI "N$(I)" £
"PP :rem 150
88 GOSUB284 :rem 142
90 NEXTI :rem 241
92 GOSUB256:PRINT"{HOME}{ 9 GIU' }{ 3 SPAZI}****
{ 2 SPAZI}SELEZIONE DI PARTENZA{ 2 SPAZI}****" :rem 47
94 FORI=2TO6 :rem 229
96 PRINT"{GIU' }{ 2 SPAZI}"I"{SIN}) = "H$(I) :rem 231
98 NEXTI :rem 249

```



```

100 PRINT"{ 2 GIU'}{ 3 SPAZI}- SI PARTE PER ";:INPUTSW           :rem 95
102 IFSW<2ORSW>6THEN92                                           :rem 147
104 PRINT"{SU}"TAB(18)H$(SW)                                     :rem 239
106 GETA$:IFA$=""THEN106                                         :rem 81
108 PRINT"{CLR}{GIU'}{YEL}{ 4 SPAZI}1 2 3 4 5 6 7 8 9 0
    { 2 SPAZI}{ 3 GIU'}{WHT}ORA SI VA PER"                       :rem 237
110 PRINT"{ 2 SU}{GRN}{ 3 SPAZI}[<A>]C[<R>]C[<R>]C[<R>]C
    [<R>]C[<R>]C[<R>]C[<R>]C[<R>]C[<R>]C[<S>]{YEL}"           :rem 127
112 FORI=0TO8                                                    :rem 12
114 PRINTI"{GRN}- - -{SPAZI}-{SPAZI}-{SPAZI}-{SPAZI}-
    {SPAZI}-{SPAZI}-{SPAZI}-{SPAZI}-"                           :rem 77
116 PRINT"{ 3 SPAZI}[<Q>]C+C+C+C+C+C+C+C+C[<W>]{YEL}"         :rem 182
118 NEXTI:PRINT"{SU}{GRN}{ 3 SPAZI}[<Z>]C[<E>]C[<E>]C
    [<E>]C[<E>]C[<E>]C[<E>]C[<E>]C[<E>]C[<E>]C[<X>}"         :rem 29
120 REM (ESTRAZIONE NUMERO CASUALE)                             :rem 160
122 DW=0:IFSW=7THEN232                                           :rem 74
124 GOSUB278                                                      :rem 184
126 X=INT(RND(1)*90)+1                                           :rem 191
128 IFX=NC(X)THEN126                                             :rem 193
130 FORI=0TO80STEP10                                             :rem 217
132 IFX<I+11THENM=3234+(80*I/10):GOTO136                       :rem 8
134 NEXTI                                                         :rem 32
136 I=I+1:M=M+2:IFX<>ITHEN136                                    :rem 224
138 I=80:NEXTI                                                  :rem 76
140 NC(X)=X:GOSUB270:POKEM,81                                    :rem 2
142 PRINT"{HOME}{ 6 GIU'}{WHT}"TAB(25)H$(SW)                   :rem 220
144 PRINT"{ 5 GIU'}{YEL}"TAB(29)"U{ 4 C}I{GIU'}{ 6 SIN}-
    { 4 SPAZI}-{GIU'}{ 6 SIN}J{ 4 C}K{SU}{ 5 SIN}{WHT}"X     :rem 182
146 REM (TEMPORIZZATORE)                                        :rem 31
148 FORTE=TC*250TO6000:NEXT                                     :rem 148
150 REM (CONTROLLO SCHEDE)                                     :rem 51
152 FORI=0TO23                                                  :rem 61
154 IFV$(NR((16*I)+1))="SV"THEN158                              :rem 99
156 GOTO200                                                       :rem 103
158 IFSW=6THEN186                                               :rem 24
160 FORO=0TO2                                                    :rem 15
162 C=0                                                         :rem 73
164 T=(16*I)+(5*O)+1                                           :rem 221
166 FORD=1TO5                                                    :rem 14
168 T=T+1                                                       :rem 224
170 IFNR(T)<>NC(NR(T))THEN174                                    :rem 216
172 C=C+1                                                        :rem 185
174 NEXTD                                                        :rem 31
176 IFC<>SWTHENNEXTO:GOTO200                                    :rem 86
178 O=2                                                         :rem 94
180 NEXTO                                                        :rem 39

```

```

182 GOTO208 :rem 110
186 C=0 :rem 79
188 FORO=1TO15 :rem 78
190 T=(16*I)+0 :rem 208
192 IFNR(T)<>NC(NR(T))THEN196 :rem 224
194 C=C+1 :rem 189
196 NEXTO :rem 46
198 IFC=15THEN206 :rem 222
200 NEXTI :rem 26
202 IFDW=1THENSW=SW+1 :rem 72
204 GOTO120 :rem 98
206 REM (RICERCA NOME VINCITORE) :rem 168
208 W2=0 :rem 144
210 FORW=1TOG :rem 41
212 FORW1=1TO3 :rem 72
214 W2=W2+1 :rem 66
216 IFNR((16*I)+1)<>CA(W2)THEN228 :rem 67
218 PRINTS$" CARTELLA"NR((16*I)+1)"DI "N$(W) " "H$(SW) :rem 57
220 NV$(SW)=NV$(SW)+N$(W)+" " :rem 11
222 W=G:W1=3 :rem 163
224 GETA$:GOSUB270:IFA$<>"C"THEN224 :rem 38
226 GOSUB286 :rem 186
228 NEXTW1,W:DW=1 :rem 41
230 GOTO200 :rem 96
232 REM (CALCOLO VINCITE) :rem 219
234 GOSUB256 :rem 182
236 PRINT"{HOME}{ 9 GIU'}{ 3 SPAZI}****{ 4 SPAZI}VINCITE :rem 62
{ 4 SPAZI}FINALI{ 4 SPAZI}****{GIU'}"
238 FORI=2TO6 :rem 21
240 P=INT(PT/10*(I/2)) :rem 184
242 IFNV$(I)=""THENDP=DP+P:P=0:GOTO246 :rem 31
244 IFI=6THENP=P+DP :rem 176
246 PRINT"{GIU'} - "H$(I) " £"P;NV$(I) :rem 62
248 NEXTI :rem 38
250 PRINT"{ 2 GIU'} C TOTALE{ 3 SPAZI}£"PT :rem 22
252 GOSUB284 :rem 183
254 RUN :rem 144
256 REM (INTESTAZIONE) :rem 111
258 PRINT"{CLR}{ 2 GIU'}{YEL}{ 3 SPAZI}[< 3 +>] :rem 111
{ 2 SPAZI}[< 3 +>]{ 2 SPAZI}[<+>] [<+>]{ 2 SPAZI}
[< 3 +>]{ 2 SPAZI}[< 3 +>]{ 2 SPAZI}[<+>]{ 4 SPAZI}
[< 3 +>]"
260 PRINT"{ 4 SPAZI}[<+>]{ 3 SPAZI}[<+>] [<+>]{ 2 SPAZI} :rem 215
[< 3 +>]{ 2 SPAZI}[<+>] [<+>]{ 2 SPAZI}[<+>] [<+>]
{ 2 SPAZI}[<+>]{ 4 SPAZI}[<+>] [<+>]"
262 PRINT"{ 4 SPAZI}[<+>]{ 3 SPAZI}[<+>] [<+>]{ 2 SPAZI} :rem 127
[<+>] [<+>]{ 2 SPAZI}[< 3 +>]{ 2 SPAZI}[<+>] [<+>]
{ 2 SPAZI}[<+>]{ 4 SPAZI}[< 3 +>]"
264 PRINT"{ 4 SPAZI}[<+>]{ 3 SPAZI}[<+>] [<+>]{ 2 SPAZI} :rem 127
[<+>] [<+>]{ 2 SPAZI}[<+>] [<+>]{ 2 SPAZI}[<+>] [<+>]

```

```

      { 2 SPAZI}[<+>]{ 4 SPAZI}[<+>] [<+>]"           :rem    53
266 PRINT"{ 4 SPAZI}[<+>]{ 3 SPAZI}[< 3 +>]{ 2 SPAZI}
      [<+>] [<+>]{ 2 SPAZI}[< 3 +>]{ 2 SPAZI}[< 3 +>]
      { 2 SPAZI}[< 3 +>]{ 2 SPAZI}[<+>] [<+>]{WHT}"
                                                    :rem   122
268 RETURN                                                    :rem   128
270 REM (SUONO)                                              :rem    98
272 X1=INT(RND(1)*500)+200                                    :rem   127
274 SOUND1,X1,Y                                             :rem   145
276 FORU=1TOY*16+30:NEXTU                                    :rem    50
278 REM (ATTESA)                                            :rem   152
280 GETA$:IFAS<>"S"THEN288                                   :rem   239
282 PRINTS$" ATTENDO..!! PER CONTINUARE PREMI (C)"
                                                    :rem   129
284 GETA$:IFAS<>"C"THEN284                                   :rem   223
286 PRINTS$"{ 38 SPAZI}"                                     :rem   232

288 RETURN                                                    :rem   130
1000 REM (NUMERI CARTELLE)                                   :rem    18
1002 DATA25,4,21,44,63,86,9,27,30,54,75,12,33,59,69,90
                                                    :rem   151
1004 DATA26,13,34,42,67,80,19,39,52,73,87,6,22,45,55,79
                                                    :rem   212
1006 DATA27,14,23,41,62,81,18,37,56,78,82,2,24,48,66,88
                                                    :rem   213
1008 DATA28,7,25,40,68,85,11,36,51,70,89,15,29,46,53,74
                                                    :rem   213
1010 DATA29,1,26,43,61,84,3,16,31,58,71,5,28,35,64,76
                                                    :rem    95
1012 DATA30,10,32,50,72,83,8,20,47,60,77,17,38,49,57,65
                                                    :rem   196
1014 DATA31,10,31,43,66,83,1,22,32,54,71,15,29,47,59,74
                                                    :rem   187
1016 DATA32,14,34,52,60,80,5,23,46,73,85,17,37,42,67,88
                                                    :rem   204
1018 DATA33,6,24,44,61,82,11,38,53,72,90,9,28,49,68,79
                                                    :rem   168
1020 DATA34,16,21,33,51,75,3,25,45,64,81,7,19,39,55,78
                                                    :rem   150
1022 DATA35,2,20,40,63,70,12,35,56,69,84,4,26,48,58,76
                                                    :rem   150
1024 DATA36,13,30,50,62,86,18,36,57,77,87,8,27,41,65,89
                                                    :rem.  221
1026 DATA61,2,26,40,64,81,9,29,46,72,87,10,39,50,68,88
                                                    :rem   165
1028 DATA62,12,30,52,73,82,8,25,42,53,90,18,36,59,67,77
                                                    :rem   211
1030 DATA63,5,21,41,66,80,17,33,54,74,85,19,24,45,69,89
                                                    :rem   212
1032 DATA64,15,32,55,71,83,3,23,37,60,75,16,38,49,58,65
                                                    :rem   209

```

```

1034 DATA65,11,31,44,61,86,1,20,34,57,70,7,27,47,63,78
:rem 148
1036 DATA66,13,22,43,62,84,4,14,35,51,76,6,28,48,56,79
:rem 162
1038 DATA67,13,33,42,63,80,5,29,47,64,72,14,37,55,76,89
:rem 221
1040 DATA68,12,35,43,61,83,1,17,38,51,75,4,25,46,69,87
:rem 158
1042 DATA69,7,22,39,53,79,15,24,40,56,67,19,28,41,59,85
:rem 223
1044 DATA70,11,32,58,71,81,6,20,48,60,88,18,30,54,65,77
:rem 203
1046 DATA71,2,21,45,62,73,3,26,49,66,82,16,34,52,78,84
:rem 159
1048 DATA72,10,31,50,70,86,8,23,44,68,90,9,27,36,57,75
:rem 162

```

C16~~+~~-plus/4

Supermind

Questo gioco per C16 e per PLUS/4 è la riproduzione di un celeberrimo gioco di logica che ha avuto il suo boom qualche anno fa: il master mind.

In esso bisogna scoprire con un numero determinato di prove (in questa versione sono disponibili 7 prove) una combinazione di 4 colori, che in questo gioco può essere scelta dal computer o da una persona.

Per mettere le pedine colorate nel

tabellone si usano i tasti da 3 a 8, e dopo aver digitato quattro colori si riceve il responso della prova con dei simboli che il computer mette a lato del rettangolo in cui si sono poste le pedine colorate.

Per ogni pedina di colore giusto nella posizione giusta il computer visualizza un tondino pieno, mentre per ogni pedina di colore giusto in una posizione sbagliata visualizza un tondino con un foro nel centro.

```

20 COLOR4,2,5:COLOR0,2,5:VOL8 :rem 223
30 PRINTCHR$(8)CHR$(142):PRINT"{BLK}{CLR}" :rem 145
40 FORX=3073TO3110:POKEX,64:NEXT :rem 229
50 FORX=3112TO3992STEP40:POKEX,93:NEXT :rem 148
60 FORX=3151TO4031STEP40:POKEX,93:NEXT :rem 137
70 FORX=4033TO4070:POKEX,64:NEXT :rem 235
80 POKE3072,85 :rem 252

```

```

90 POKE3111,73:POKE4032,74:POKE4071,75 :rem 138
100 PRINT"[<7>]{ 3 GIU'}{ 4 DES}{RVS}{FLASH ON}
      { 3 SPAZI}S{ 2 SPAZI}U{ 2 SPAZI}P{ 2 SPAZI}E
      { 2 SPAZI}R{ 3 SPAZI}M{ 2 SPAZI}I{ 2 SPAZI}N
      { 2 SPAZI}D{ 3 SPAZI}" :rem 238
110 DIMA%(4):DIMB%(4):DIMD%(4) :rem 89
120 D$="{ 10 SPAZI}ISTRUZIONI (S/N)?{ 9 SPAZI}" :rem 252
130 PRINT"{ 2 GIU'}" :rem 135
140 PRINT"{SU}{ 2 DES} [<7>";D$:FORX=1TO79:NEXT :rem 225
150 D$=D$+LEFT$(D$,1):D$=MID$(D$,2,36) :rem 188
160 GETA$:IFA$<>"N"ANDA$<>"S"THEN140 :rem 35
170 GOSUB600 :rem 174
180 PRINT"{HOME}{GIU'}"SPC(18)"C) COMPUTER"CHR$(13)SPC(18
      )"{GIU'}A) AVVERSARIO" :rem 100
190 GETA$:IFA$<>"C"ANDA$<>"A"THEN190 :rem 14
200 PRINT"{HOME}{GIU'}"SPC(18)"{ 11 SPAZI}"CHR$(13)SPC(18
      )"{GIU'}{ 20 SPAZI}" :rem 22
210 IFA$="C"THENGOSUB790:GOTO280 :rem 167
220 R=1 :rem 84
230 FORX=1TO4 :rem 25
240 GETA$:IFA$=CHR$(20)ANDX>1THENX=X-1:R=R-2:POKE3115+R,1
      60 :rem 251
250 IFVAL(A$)<3ORVAL(A$)>8THEN240 :rem 163
260 A%(X)=VAL(A$):POKE3115+R,81 :rem 194
270 R=R+2:NEXT :rem 80
280 GOSUB770 :rem 184
290 PRINT"{HOME}"SPC(32)"{ 5 GIU'}{RED}3=Q{ 3 GIU'}
      { 3 SIN}{CYN}4=Q{ 3 GIU'}{ 3 SIN}{PUR}5=Q{ 3 GIU'}
      { 3 SIN}{GRN}6=Q{ 3 GIU'}{ 3 SIN}{BLU}7=Q{ 3 GIU'}
      { 3 SIN}{YEL}8=Q{BLU}" :rem 204
300 Y=840:R=1 :rem 191
320 SOUND2,19,19 :rem 124
330 FORX=1TO1000:NEXT :rem 32
350 FORX=1TO4:D%(X)=A%(X):NEXT :rem 45
360 FORX=1TO4 :rem 29
370 GETA$:IFA$=CHR$(20)ANDX>1THENX=X-1:R=R-2:POKE3115+R+Y
      ,32 :rem 81
380 IFVAL(A$)<3ORVAL(A$)>8THEN370 :rem 171
390 B%(X)=VAL(A$):POKE3115+R+Y,81:POKE2091+R+Y,B%(X)+63
      :rem 81
400 R=R+2:NEXT :rem 75
410 FORX=1TO4:IFB%(X)=D%(X)THENG$=G$+"Q":D%(X)=255:B%(X)=
      254 :rem 10
420 NEXT :rem 213
430 IFD%(1)=B%(2)THENG$=G$+"W":D%(1)=255:B%(2)=254
      :rem 186
440 IFD%(1)=B%(3)THENG$=G$+"W":D%(1)=255:B%(3)=254
      :rem 189
450 IFD%(1)=B%(4)THENG$=G$+"W":D%(1)=255:B%(4)=254
      :rem 192

```

```

460 IFD%(2)=B%(1) THENG$=G$+"W":D%(2)=255:B%(1)=254
:rem 189
470 IFD%(2)=B%(3) THENG$=G$+"W":D%(2)=255:B%(3)=254
:rem 194
480 IFD%(2)=B%(4) THENG$=G$+"W":D%(2)=255:B%(4)=254
:rem 197
490 IFD%(3)=B%(1) THENG$=G$+"W":D%(3)=255:B%(1)=254
:rem 194
500 IFD%(3)=B%(2) THENG$=G$+"W":D%(3)=255:B%(2)=254
:rem 188
510 IFD%(3)=B%(4) THENG$=G$+"W":D%(3)=255:B%(4)=254
:rem 193
520 IFD%(4)=B%(1) THENG$=G$+"W":D%(4)=255:B%(1)=254
:rem 190
530 IFD%(4)=B%(2) THENG$=G$+"W":D%(4)=255:B%(2)=254
:rem 193
540 IFD%(4)=B%(3) THENG$=G$+"W":D%(4)=255:B%(3)=254
:rem 196
550 PRINT"{HOME}":FORX=1TO((Y-40)/40):PRINT"{GIU' }";NEXT
:PRINTSPC(18)"{GRN}"G$ :rem 43
560 IFG$="{ 4 Q}" THEND$="SI":GOTO820 :rem 27
570 IFY=120 THEND$="NO":GOTO820 :rem 22
580 Y=Y-120:G$="":R=1 :rem 108
590 GOTO320 :rem 108
600 IFA$="N" THEN700 :rem 31
610 PRINT"{CLR}{ 2 GIU' }[<7>]{RVS}{ 15 SPAZI}ISTRUZIONI
{ 15 SPAZI}" :rem 233
620 PRINT"{ 2 GIU' } IN QUESTO GIOCO SI PUO' SFIDARE IL C1
6{ 10 SPAZI}O UN AVVERSARIO UMANO." :rem 10
630 PRINT"{GIU' }{ 2 SPAZI}SI HANNO {GRN}7[<7>] TENTATIVI
A DISPOSIZIONE." :rem 65
640 PRINT"{GIU' }PER LE PEDINE USARE I TASTI {RED}3 {CYN}4
{PUR}5 {GRN}6 {BLU}7 {YEL}8[<7>]" :rem 4
650 PRINT"{GIU' } PER CORREGGERE GLI ERRORI PREMERE {RVS}D
EL" :rem 239

660 PRINT"{GIU' }{GRN} Q[<7>]{ 2 SPAZI}INDICA GIUSTO COLOR
E-POSTO GIUSTO" :rem 235
670 PRINT"{GIU' }{GRN} W[<7>]{ 2 SPAZI}INDICA GIUSTO COLOR
E-POSTO SBAGLIATO" :rem 173
680 PRINT"{RVS}{ 3 GIU' }{ 7 SPAZI}PREMI RETURN{ 2 SPAZI}
PER INIZIARE{ 7 SPAZI}" :rem 83
690 GETA$:IFA$<>CHR$(13) THEN690 :rem 16
700 PRINT"{CLR}"; :rem 54
710 FORX=1TO8 :rem 32
720 PRINT"[<7>]{ 3 SPAZI}O[< 7 Y>]P{ 31 SPAZI}[<H>]
{ 7 SPAZI}[<N>]" :rem 2
730 PRINT"{ 3 SPAZI}[<H>]{ 7 SPAZI}[<N>]" :rem 201
740 NEXT :rem 218
750 PRINT"{ 3 SPAZI}[< 9 Y>"; :rem 23

```

```

760 PRINT"{HOME}{ 3 GIU'}S{ 2 GIU'}{SIN}U{ 2 GIU'}{SIN}P
    { 2 GIU'}{SIN}E{ 2 GIU'}{SIN}R{ 2 GIU'}{SIN}M
    { 2 GIU'}{SIN}I{ 2 GIU'}{SIN}N{ 2 GIU'}{SIN}D"
                                                    :rem 99
770 FORX=1TO9:FORY=0TO2:POKE3074+X+Y*40,160:POKE2069+X+Y*
    40,14:NEXTY,X
                                                    :rem 40
780 RETURN
                                                    :rem 127
790 FORX=1TO4
                                                    :rem 36
800 FORY=1TO3:E=RND(1)*5+3:NEXTY
                                                    :rem 205
810 A%(X)=E:NEXT:RETURN
                                                    :rem 189
820 FORX=1TO4:POKE3116+C,81:POKE2092+C,A%(X)+63:C=C+2:NEX
    T
                                                    :rem 155
830 PRINT"{HOME}";:IFD$="NO"THENPRINTSPC(16)"NON HAI INDO
    VINATO."
                                                    :rem 141
840 IFD$="SI"THENPRINTSPC(16)"BRAVO, HAI INDOVINATO!"
                                                    :rem 226
850 PRINT"{HOME}{ 2 GIU'}"SPC(16)"GIOCHI ANCORA (S/N)?"
                                                    :rem 76
860 GETA$:IFA$<>"N"ANDA$<>"S"THEN860
                                                    :rem 51
870 IFA$="N"THENSCNCLR:END
                                                    :rem 103
880 RUN
                                                    :rem 149
                                                    :rem 0

```



Semaforo

È un gioco per due persone scritto appositamente per controllare ed allenare i riflessi.

All'inizio viene chiesto ai giocatori se desiderano leggere le istruzioni esplicative comprese nel programma. Dopo le istruzioni inizia il gioco vero e proprio e un semaforo viene visualizzato al centro dello schermo. Le luci di questo semaforo cominciano ad illuminarsi ad una ad una dall'alto verso il basso, come in un conto alla rovescia. Quando l'ultima luce si accende, può apparire in tre diversi colori. I due giocatori, per partecipare al gioco, devono usare due tasti: il giocatore di sinistra il tasto "A", quello di

destra il tasto "=". I vari colori che la lampada finale del semaforo può assumere sono tre: azzurro, bianco o viola. Nel caso la lampada diventi bianca, il primo dei due giocatori che preme il suo tasto vince la manche e si aggiudica un punto. Se invece la lampada diventa azzurra, solo il giocatore di sinistra ha diritto a premere il tasto, guadagnando un punto. Nel caso che la lampada divenga viola, tocca al giocatore di destra premere il suo tasto. Se nella troppa foga un giocatore preme il proprio bottone quando non è il suo turno, perde tre punti. Il primo che raggiunge i 10 punti vince la partita.

```

15 POKE53281,15:POKE53280,2:F=RND(0) :rem 21
20 A1=55475:A2=A1+40:A3=A2+40:A4=A3+40:A5=A4+40:A6=A5+40:
RS=0:LS=0:DE=90 :rem 193
30 PRINT"{CLR}{GIU'}{BLK}{ 9 DES}ISTRUZIONI?(S=SI/N=NO)"
:rem 160
35 GETG$:IFG$="S"THEN350 :rem 130
40 IFG$<>"N"THEN35 :rem 1
45 PRINTCHR$(142)"{CLR}{GIU'}{BLU}{ 12 DES}S E M A F O R
O{BLK}" :rem 46
50 PRINT"{GIU'}"TAB(19)"[<O>]" :rem 214
55 PRINTTAB(18)"[<L>]Q[<J>]" :rem 8
57 PRINTTAB(18)"[<L>]Q[<J>]" :rem 10
60 PRINTTAB(18)"[<L>]Q[<J>]" :rem 4
62 PRINTTAB(18)"[<L>]Q[<J>]" :rem 6
65 PRINTTAB(18)"[<L>]Q[<J>]" :rem 9
67 PRINTTAB(18)"[<L>]Q[<J>]" :rem 11
70 PRINTTAB(19)"{RVS}{OFF}" :rem 110
75 PRINTTAB(19)"[<+>]" :rem 117
80 PRINTTAB(19)"[<+>]" :rem 113
85 PRINTTAB(19)"[<+>]" :rem 118
90 PRINTTAB(19)"[<+>]" :rem 114
95 PRINTTAB(18)"[< 3 U>]{GIU'}" :rem 9
100 POKEA1,2 :rem 144
105 FORF=1TO2000:NEXT :rem 15
110 POKEA1,0:POKEA2,2:GOSUB500 :rem 24
117 POKEA2,0:POKEA3,7:GOSUB500 :rem 38
118 POKEA3,0:POKEA4,7:GOSUB500 :rem 41
119 POKEA4,0:POKEA5,6:GOSUB500:POKE198,0 :rem 146
120 D=INT(RND(1)*4)+1 :rem 112
125 IFD=1THENE=1 :rem 187
130 IFD=2THENE=1 :rem 184
135 IFD=3THENE=4 :rem 193
137 IFD=4THENE=3 :rem 195
140 POKEA5,0:POKEA6,E:POKES,235 :rem 105
145 ONDGOTO150,150,165,165 :rem 156
150 GETG$:IFG$="A"ORG$="="THEN160 :rem 103
155 GOTO150 :rem 106
160 POKES,0:GOTO260 :rem 128
165 GETG$:IFG$="A"ORG$="="THEN175 :rem 115
170 GOTO165 :rem 109
175 IFG$="A"ANDE=3THEN260 :rem 168
180 IFG$="="ANDE=4THEN260 :rem 161
190 IFE=3THENGOSUB250 :rem 36
192 IFDE>0THENDE=DE-50 :rem 50
195 PRINT"{RVS}SBAGLIATO{OFF}" :rem 170
200 IFE=3THENGOSUB250 :rem 28
205 PRINT"{GIU'}NON E' IL" :rem 101
210 IFE=3THENGOSUB250 :rem 29
215 PRINT"TUO TURNO." :rem 39
220 IFE=3THENRS=RS-3:GOTO230 :rem 246
225 LS=LS-3 :rem 116

```



```

230 REM :rem 121
238 FORF=1TO1000:NEXTF :rem 91
240 POKE198,0:GOTO45 :rem 159
250 PRINTTAB(30); :rem 234
255 RETURN :rem 124
260 IFG$="="THEN290 :rem 26
265 LS=LS+1:DE=DE-50:IFLS=10THEN400 :rem 88
270 PRINT"BRAVO !!!!!" :rem 104
275 PRINT"{GIU'}PUNTI{DES}"LS :rem 204
277 REM :rem 132
282 FORF=1TO2000:NEXTF :rem 91
285 POKE198,0:GOTO45 :rem 168
290 RS=RS+1:DE=DE-50:IFRS=10THEN400 :rem 104
295 PRINTTAB(30);"BRAVO !!!!!" :rem 53
300 PRINTTAB(30);"PUNTI{DES}"RS :rem 124
301 GOTO277 :rem 109
350 PRINTCHR$(14)"{CLR}{BLU}{ 12 DES}S E M A F O R O" :rem 138
355 PRINT"{BLK}{GIU'}E' UN GIOCO PER PROVARE I RIFLESSI." :rem 237
357 PRINT"IL GIOCATORE DI SINISTRA USA IL TASTO":rem 37
358 PRINT"A, QUELLO DI DESTRA USA IL TASTO '='." :rem 144
360 PRINT"SE L'ULTIMA LUCE AD ACCENDERSI E' BIANCA"; :rem 133
362 PRINT"IL PRIMO GIOCATORE CHE PREME IL SUO TA- STO GUA :rem 15
DAGNERA' UN PUNTO."
364 PRINT"SE L'ULTIMA LUCE E' AZZURRA IL GIOCATORE"; :rem 246
365 PRINT"DI SINISTRA DEVE PREMERE IL SUO TASTO;" :rem 239
366 PRINT"SE E' VIOLA TOCCA INVECE AL GIOCATORE :rem 187
{ 3 SPAZI}DI DESTRA."
367 PRINT"{GIU'}**{ 2 SPAZI}PREMERE UN TASTO PER CONTINUA :rem 48
RE.{ 2 SPAZI}**";
368 GETG$:IFG$=""THEN368 :rem 113
371 PRINT"{CLR}{BLU}{ 12 DES}S E M A F O R O{BLK}" :rem 102
373 PRINT"{GIU'}SE UN GIOCATORE PREME IL SUO TASTO QUAN- :rem 4
;
374 PRINT"DO E' LA VOLTA DELL'AVVERSARIO, PERDE 3 PUNTI." :rem 88
375 PRINT"IL PRIMO CONCORRENTE CHE RAGGIUNGE I 10 PUNTI V :rem 148
INCE."
378 PRINT"{GIU'}**{ 2 SPAZI}PREMERE UN TASTO PER CONTINUA :rem 50
RE.{ 2 SPAZI}**";
379 GETT$:IFT$=""THEN379 :rem 143
380 PRINTCHR$(142)"{CLR}":GOTO45 :rem 196
400 IFRS=10THENGOSUB250 :rem 172
405 PRINT"{RVS}HAI VINTO{OFF}" :rem 112
410 IFRS=10THENGOSUB250 :rem 173

```

```

415 PRINT" {GIU' }BRAVISSIMO" :rem 123
420 REM :rem 122
430 PRINT" {RVS}{ 11 DES}GIOCHI ANCORA (S/N) {OFF}":rem 211
435 GETG$:IFG$="S"THEN15 :rem 132
440 IFG$="N"THENPRINTCHR$(142)" {CLR}":END :rem 238
445 GOTO435 :rem 114
500 POKES,231:FORF=1TO300:NEXT:POKES,0 :rem 117
510 FORF=1TODE:NEXT:RETURN :rem 240

```

New York



Questo originale e divertente gioco vi permette di osservare New York da un elicottero e di regolarne il traffico. All'inizio vedrete tutte le strade deserte, ma poco per volta esse si popoleranno di automobili. Il vostro compito sarà quello di dirigere efficacemente il caotico traffico della metropoli. Quando un'automobile, rappresentata da un rettangolino nero, raggiunge uno dei semafori, raffigurati invece da lettere e numeri, si ferma se il semaforo è rosso, oppure attraversa l'incrocio se il semaforo è verde. Per modificare il colore del semaforo dovete premere il tasto corri-

spondente al simbolo che lo rappresenta.

All'inizio del gioco avete 100 punti. Guadagnate 10 punti ogni volta che un veicolo attraversa un incrocio e 25 punti se un'automobile attraversa tutta la città. Se due vetture si scontrano ad un incrocio perdetevi 100 punti e la strada rimane temporaneamente bloccata. Il vostro punteggio diminuisce anche in proporzione al numero di auto ferme in coda ai semafori. Se restate con zero punti la partita ha termine, altrimenti il gioco dura 5 minuti. Più punti otterrete, più brillante sarà la vostra carriera di vigile!

```

10 REM NEW YORK :rem 116
15 POKE53281,7:POKE53280,11 :rem 244
20 A=RND(-TI):TI$="000000":TB=TI:PU=100 :rem 210
30 PRINT" {CLR} [<7>] [< 6 +>] _ [< 22 +>] _ [< 6 +>]"; :rem 98
40 PRINT" [< 6 +>]A_ [< 22 +>]B_ [< 6 +>]"; :rem 185
50 PRINT" { 7 SPAZI}_ C { 5 SPAZI} [<R>] D { 5 SPAZI} [<R>] E :rem 161
   { 6 SPAZI}_ F { 5 SPAZI}";
60 PRINT" { 7 C}_+ { 7 C}_+ { 7 C}_+ { 8 C}_+ { 7 C}"; :rem 74

```

```

70 PRINT"{ 5 SPAZI}G _{ 5 SPAZI}H _{ 5 SPÄZI}I _
   { 6 SPAZI}J _{ 7 SPÄZI}"; :rem 9
80 PRINT"< 6 +>] _K[< 5 +>] _L[< 5 +>] _M[< 6 +>] _N
   [< 6 +>]"; :rem 66
90 A$=" [< 6 +>] _ [< 5 +>] _ [< 5 +>] _ [< 6 +>] _
   [< 6 +>]":PRINTA$ A$ A$; :rem 28
100 PRINT"< 6 +>]O_ [< 5 +>]P_ [< 5 +>]Q_ [< 6 +>]R_
   [< 6 +>]"; :rem 123
110 PRINT"< 6 +>] _ S{ 5 SPAZI}_ T{ 5 SPAZI}_ U
   { 6 SPAZI}_ [< 6 +>]"; :rem 214
120 PRINT"< 6 +>][<Q>]+{ 7 C}+{ 7 C}+{ 8 C}+ [<W>]
   [< 6 +>]"; :rem 243
130 PRINT"< 6 +>] _{ 5 SPAZI}V _{ 5 SPAZI}W _{ 6 SPAZI}
   X _ [< 6 +>]"; :rem 225
140 PRINT"< 6 +>] _Y[< 5 +>] _Z[< 5 +>] _0[< 6 +>] _1
   [< 6 +>]"; :rem 81
150 PRINTA$ A$ A$; :rem 141
160 PRINT"< 6 +>]2_ [< 5 +>]3_ [< 5 +>]4_ [< 6 +>]5_
   [< 6 +>]"; :rem 13
170 PRINT"{ 7 SPAZI}_ 6{ 5 SPAZI}_ 7{ 5 SPAZI}_ 8
   { 6 SPAZI}_ 9{ 5 SPAZI}"; :rem 246
180 PRINT"{ 7 C}+{ 7 C}+{ 7 C}+{ 8 C}+{ 7 C}"; :rem 125
190 PRINT"{ 5 SPÄZI}+ _{ 5 SPÄZI}E [ <E> ]{ 5 SPAZI}@ [ <E> ]
   { 6 SPAZI}* _{ 7 SPÄZI}"; :rem 179
200 PRINT"< 6 +>] _↑[< 22 +>] _=[< 6 +>]"; :rem 255
210 PRINT"< 6 +>] _ [< 22 +>] _ [< 6 +>]"; :rem 101
300 PRINT"< 6 +>] _ [< 22 +>] _ [< 6 +>]{HOME}";
   :rem 120

400 AW=40:SC%=1023:CL=55295 :rem 213
500 DIMX%(AW),Y%(AW),R%(AW),L%(41),A%(3),F%(9) :rem 126
510 FORI=0TO41:READL%(I):NEXT :rem 21
520 DATA47,72,90,98,106,115,166,174,182,191,209,217,225,2
   34,367,375,383,392 :rem 202
530 DATA410,418,426,494,502,511,529,537,545,554,687,695,7
   03,712 :rem 131
535 DATA730,738,746,755 :rem 185
540 DATA806,814,822,831,849,874 :rem 74
550 FORI=0TO41:A=PEEK(L%(I)+SC%):POKEL%(I)+SC%,AOR128
   :rem 97

560 POKEL%(I)+CL,2:NEXT :rem 102
590 A%(0)=1:A%(1)=-1:A%(2)=40:A%(3)=-40 :rem 102
600 DEFNR(X)=INT(RND(1)*X) :rem 85
1000 FORI=0TOAW:R%=R%(I):IFR%=0THEN1099 :rem 71
1010 X%=X%(I):Y%=Y%(I) :rem 160
1020 IFR%AND1THENX%=X%+1:F%=98:O0%=226:O%=64:S%=40:IFX%>4
   0THEN1200 :rem 25
1030 IFR%AND2THENX%=X%-1:F%=226:O0%=98:O%=64:S%=-40:IFX%<
   1THEN1200 :rem 21
1040 IFR%AND4THENY%=Y%+1:F%=97:O0%=225:O%=93:S%=-1:IFY%>2
   3THEN1200 :rem 28

```

```

1050 IFR%AND8THENY%=Y%-1:F%=225:O%=97:O%=93:S%=1:IFY%<0T
HEN1200 :rem 191
1060 IFR%AND16THEN1100 :rem 248
1070 IFR%AND32THEN1400 :rem 250
1080 IFR%AND64THENO%=91:R%=R%-64:R%(I)=R%:PU=PU+10
:rem 158
1090 GOTO1300 :rem 199
1099 NEXT:GOTO2000 :rem 71
1100 IF(PEEK(X%(I)+40*Y%(I)+S%+CL)AND15)<>2THENR%(I)=(R%O
R32)ANDNOT16:GOTO1300 :rem 239
1110 PU=PU-1:GOTO1099 :rem 238
1200 PU=PU+25:PN=PN+1:R%(I)=0:P%=-1:GOTO1330 :rem 39
1300 P%=X%+40*Y%:PS%=PEEK(P%+SC%):IFPS%=F%ORPS%=160THENPU
=PU-1:GOTO1099 :rem 10
1310 A=PEEK(P%+S%+SC%):IFA<>32THENR%(I)=R%OR16 :rem 156
1320 IFPS%=00%THENF%=160 :rem 108
1325 V%=PEEK(X%(I)+40*Y%(I)+SC%):IFV%=160THENO%=00%
:rem 99
1327 IFV%=102THENR%(I)=0:AO=AO+1:PU=PU-100:GOTO1099
:rem 12
1330 POKEP%+SC%,F%:POKEP%+CL,0 :rem 220
1335 IFO%=64ORO%=93ORO%=91THENPOKEX%(I)+40*Y%(I)+CL,14
:rem 129
1340 POKEX%(I)+40*Y%(I)+SC%,O%:X%(I)=X%:Y%(I)=Y%:GOTO1099
:rem 195
1400 P%=X%+40*Y%:A=PEEK(P%+SC%):IFA=F%THEN1099 :rem 104
1410 IFA<>91THEN1500 :rem 111
1420 VB=0:FORJ=0TO3:A=2↑J:B=PEEK(P%+SC%+A%(J)):IFNOT(B=64
ORB=93)THENVB=VBORA :rem 192
1430 NEXT :rem 7
1440 B=2↑FNR(4):IF(BANDVB)=0THENR%(I)=BOR64:GOTO1330
:rem 123
1450 IFVB=15THEN1099 :rem 150
1460 GOTO1440 :rem 205
1500 AO=AO+1:R%(I)=0:IFA<>102THENF%(AF)=P%:AF=AF+1:PU=PU-
100 :rem 210
1510 F%=102:GOTO1330 :rem 60
2000 IFR%(TL)=0THEN2100 :rem 24
2010 TL=TL+1:IFTL>AWTHENTL=0 :rem 23
2020 GOTO3000 :rem 192
2100 A=FNR(4):R%(TL)=2↑A:ONA+1GOTO2110,2120,2130,2140
:rem 163
2110 X%(TL)=1:GOTO2200 :rem 215
2120 X%(TL)=40:GOTO2200 :rem 11
2130 Y%(TL)=0:GOTO2300 :rem 218
2140 Y%(TL)=23:GOTO2300 :rem 16
2200 IFFNR(2)=1THENY%(TL)=3:GOTO3000 :rem 110
2210 Y%(TL)=19:GOTO3000 :rem 17
2300 IFFNR(2)=1THENX%(TL)=8:GOTO3000 :rem 115
2310 X%(TL)=33:GOTO3000 :rem 13

```


```

3000 GETA$: IFA$="" THEN 3090 :rem 178
3010 A=ASC(A$): IFA>64 AND A<91 THEN A=A-65: GOTO 3100 :rem 248
3020 IFA>47 AND A<58 THEN A=A-22: GOTO 3100 :rem 177
3030 IFA=43 THEN A=36: GOTO 3100 :rem 135
3040 IFA=92 THEN A=37: GOTO 3100 :rem 141
3050 IFA=64 THEN A=38: GOTO 3100 :rem 142
3060 IFA=42 THEN A=39: GOTO 3100 :rem 140
3070 IFA=94 THEN A=40: GOTO 3100 :rem 140
3080 IFA=61 THEN A=41: GOTO 3100 :rem 136
3090 GOTO 4000 :rem 201
3100 P%=L%(A): A=PEEK(P%+CL): IF (A AND 15)=2 THEN A=5: GOTO 3110 :rem 243
3105 A=2 :rem 121
3110 POKE P%+CL, A :rem 144
3120 GOTO 3000 :rem 194
4000 REM :rem 168
4010 PRINT "{HOME}{ 22 GIU' } [<4>] "SPC(18)MID$(TI$,4,1) ": "R
      IGH T$(TI$,2) "{HOME}" :rem 10
4020 IF PU<0 THEN PU=0 :rem 167
4030 PRINT "{HOME}{RVS} [<4>] "SPC(18)RIGHT$(" { 3 SPAZI } "+ST
      R$(PU),4) :rem 201
4040 IF PU=0 THEN 9000 :rem 97
4100 IF FNR(25) OR F%(0)=0 THEN 4200 :rem 225
4110 POKE F%(0)+SC%,91: FOR I=0 TO 8: F%(I)=F%(I+1): NEXT I: AF=AF-
      1 :rem 72
4200 IF TI-TB>18E3 THEN 9000 :rem 204
5000 GOTO 1000 :rem 191
9000 PRINT "{HOME}{ 7 GIU' } {RVS} [<4>] { 6 DES } { 4 SPAZI } F I
      N E { 4 SPAZI } G I O C O { 4 SPAZI} ": FOR I=1 TO 4000: NEXT
      I :rem 198
9010 IF PU>0 THEN PRINT "{CLR}{GIU'} {BLK} CE L'HAI FATTA! HAI
      OTTENUTO"; PU; "PUNTI": GOTO 9030 :rem 128
9020 PRINT "{CLR}{GIU'} {BLK} PESSIMO! NON SEI RIUSCITO A D
      IRIGERE" :rem 219
9025 PRINT " IL TRAFFICO NEANCHE PER 5 MINUTI." :rem 150
9030 PRINT "{ 2 GIU' } {RVS} {BLU} { 9 DES } GIOCHI ANCORA (S/N
      )? " :rem 188
9040 GETA$: IFA$="S" THEN RUN :rem 56
9050 IFA$="N" THEN PRINT "{CLR}": END :rem 53
9080 GOTO 9040 :rem 215

```

FINALMENTE!

La Softrivista che ti gasa!



**QUALCOSA DI SUPER, DI INEDITO,
DI IRRESISTIBILE**

IL VERO GIOCO COMINCIA ADESSO

IN EDICOLA JACKSON SOFT SERIE ORO

I giochi esclusivi per
Commodore 64 e Spectrum 48 K
importati dall'Inghilterra, mai
presentati in Italia.
Una sfida Jackson al già visto, al
già fatto, al... già registrato.



La prima
puntata del
fantastico,
inedito
PYJAMARAMA

Corri in edicola, il vero gioco comincia solo adesso
e se sei davvero bravo partecipa alla "sfida al campione",
utilizzando il tagliando che troverai sull'ultima pagina
di copertina di ogni numero.



GRUPPO
EDITORIALE
JACKSON

IN EDICOLA

SUPER

SINC

SUPER

COMMODORE

N°7



**SuperSinc e
Super Commodore
sono idee**

J.soft EDITRICE

**CON
CASSETTA**